Slide 1

# Modelling Informative Dropout using NONMEM

Nick Holford
University of Auckland



Slide 2

## General Principles (Abowd)

- The basic insight is that missing data should be modeled using the same probability and statistical tools that are the basis of all data analysis
- Missing data are not an anomaly to be swept under the carpet
- They are an integral part of every analysis

From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8 (no longer available)



Slide 3

## Missing Data Mechanisms

Little & Rubin distinguish between:

- Missing Completely at Random (MCAR)
- Missing at Random (MAR)
- Not Missing at Random (NMAR)

*Statistical Analysis with Missing Data*, 2nd edition, Roderick J. A. Little and Donald B. Rubin (New York: John Wiley & Sons, 2002).



When data are missing there are 3 commonly recognized categories. The categories are based on a mechanism (M) for causing data to be missing. The probability of data being missing may be

- independent of any observed value (missing completely at random)
- Predictable from an observed value (missing at random)
- Predictable from the unobserved value i.e. the missing data (not missing at random)

When data is not missing at random it is also called 'informative missingness'

| | | |
|---|---|---|
| Slide 4 | **Shafer's Taxonomy of Methods**<br>  <br>Shafer JL http://www.stat.psu.edu/%7Ejls/asa97/slide7.html (no longer available)<br> | Shafer has identified a series of methods that have been developed for analysing missing data. NONMEM provides a likelihood based approach. |
| Slide 5 | **Modelling Informative Missingness**<br>Model for the Data (PKPD and Disease Progress)<br>+<br>Model for the Missingness Mechanism<br>= Maximum Likelihood Joint Model<br> | Models for 'not missing at random' (or 'informative missingness') depend on using a model to predict the unobserved values that are expected to determine why data is missing. This model for the data is simply a pharmacokinetic-pharmacodynamic plus disease progress model.<br><br>The additional model required to predict missing data is a model for the missingness mechanism.<br><br>The combination of the model for the data with the model for missingness forms a joint model. Maximum likelihood estimation can be used to determine the parameters of the join model. |
| Slide 6 | **Disease Progress Model**<br>  <br>Fig. 1. Observed viral load over time for a random sample of 40 patients. The thick lines are loess smooth of the data.<br>$Y_{ij} = a_{trt} + \eta_{1i} + (b_{trt} + \eta_{2i}) * t_j + \varepsilon_{ij}$<br>$a_{trt}$ and $b_{trt}$ are fixed effect parameters which may depend on treatment covariate (trt)<br>Model can describe symptomatic (trt affects a) and protective (trt affects b) disease progress actions of treatment<br>Hu C, Sale ME. A joint model for nonlinear longitudinal data with informative dropout. J Pharmacokinet Pharmacodyn 2003;30(1):83-103<br> | This figure shows an example of viral load observations in patients with human immunodeficiency virus (HIV) infection. The left hand panel is for patients treated with amprenavir and the right hand panel for patients treated with indinavir.<br><br>A linear disease progress model might be used to describe the time course of viral load. The model for the data has two fixed effect parameters (intercept (a) and slope (b)). These parameters may be affected by a PKPD drug action model for the effect of treatment. Each parameter has an associated random effect to account for between subject variability. There is a residual error parameter to describe measurement error and model misspecification. |

Slide 7

## Disease Progression and Last Observation Carried Forward Statistical Madness






Slide 8

## Better Models for Missing Data

CLINICAL PHARMACOLOGY & THERAPEUTICS
VOLUME 72 NUMBER 6 DECEMBER 2002

COMMENTARY

More efficient clinical trials through use of scientific model-based statistical tests

E. Niclas Jonsson, PhD, and Lewis B. Sheiner, MD *Uppsala, Sweden, and San Francisco, Calif*



Slide 9

## Hu & Sale Terminology

Variables are dropout time *T*, observed ($Y_O$) and unobserved values ($Y_U$) of disease progress state (e.g. HIV viral load)

**(a) completely random** (CRD), if *Ti* is independent of η, and therefore ($Y_O$, $Y_U$);

(b) **random** (RD), if *Ti* (given $Y_O$) is independent of $Y_U$, but may depend on $Y_O$. In addition, any dependence of *Ti* on η is only through $Y_O$;

(c) **informative** (ID), if *Ti* (given $Y_O$) depends on $Y_U$, or explicitly depends on η other than through $Y_O$.

Hu C, Sale ME. A joint model for nonlinear longitudinal data with informative dropout. J Pharmacokinet Pharmacodyn 2003;30(1):83-103



Hu & Sale have described how to model missing data and have proposed a modified terminology for missingness mechanism. The three categories are essentially the same as those defined by Little & Rubin but the names are a shorter.

They illustrate their terminology with the example of predicting the probability of a patient dropping out of a trial given the time of dropout and observed and unobserved values of viral load. The unobserved values are predicted from a PKPD and disease progress model.

| | | |
|---|---|---|
| Slide 10 |  | The probability of a dropping out at a particular time can be predicted by describing the hazard for dropout. Hazard is the instantaneous risk of dropping out. As time passes the cumulative hazard predicts the risk of dropping out at a particular time.<br>The hazard model shown here is a proportional hazard models which is widely used for survival analysis. Beta0 is the baseline hazard. Beta is a parameter describing how the hazard depends on some other value, X.<br>The risk is the cumulative hazard. It is obtained by integrating hazard with respect to time.<br>The probability of survival (not dropping out) can be predicted from the risk by assuming an exponential distribution for not dropping out.<br>The dropout probability can be predicted from the survival probability. |
| Slide 11 |  | Using NONMEM nomenclature, DV is a the observed drop-out state. If it is 0 it means the subject has not dropped out. If it is 1 then the subject has dropped out at some time between the last observation when they had not dropped out and the current time when it is known they have dropped out.<br>The probability of a drop-out event is conditional on whether drop-out has not occurred at the time of the probability prediction. This model shows the probability of not dropping out depends on the hazard accumulated from time 0 until the time of the current observation given that the subject has not dropped out (DV=0). In this example the hazard uses the predicted disease state value rather than the observed disease state value.<br>At the time a subject is known to have dropped out (DV=1) the probability of this happening is the product of the probability they the subject had not dropped out at the time of the last observation (when it was known they had not dropped out) and the probability of dropping out at some time between the last observation and the current time when it known the subject has dropped out. |

Slide 12

## -2 Log Likelihood

Disease Progress

$$ELS = \sum_{i=1}^{NSUB} \sum_{j=1}^{NOBS} \left( \frac{\left(Y_{OBS,ij} - Y_{PRED,ij}\right)^2}{Var_{ij}} + \ln\left(Var_{ij}\right) \right)$$

$$CCONTR_{-2LL,ij} = \left( \frac{Y_{OBS,ij} - Y_{PRED,ij}}{SD_{ij}} \right)^2 + 2 \cdot \ln\left(SD_{ij}\right)$$

Dropout Probability

$$\text{CCONTR}_{-2\text{LL}} = -2 \cdot \ln(\text{Like})$$



The joint model for the likelihood depends on calculating the likelihood for the disease state observation (e.g. viral load) and the likelihood for the drop-out observation. The disease state observation is usually a continuous variable and the likelihood can be calculated using the extended least squares method. The likelihood of the drop-out observation is simply the probability of the observation (Pr). In this example the value that is computed is -2 times the log likelihood. This is the usual way that NONMEM computes it's objective function.
The contribution from each observation of disease state or drop-out is computed by a function called CCONTR in NONMEM. The equations shown here show the values computed by CCONTR using -2 time the log likelihood contribution.

Slide 13

## Hu & Sale Code

```
$MODEL COMP=CUMHAZ ; compartment for integration of hazard
COMP=(HZLAST, INITIALOFF) ; comp for LAST PERIOD hazard
$PK
INTERC=(THETA(1) - THETA(2)*(TRT-1))+ETA(1)
SLOPE=THETA(3)+ETA(2)
BSHZ=THETA(4)
BETA=THETA(5)
BET2=THETA(6)
$DES
VIRL=INTERC+SLOPE*(T-12)
TEMP=BETA*LOCF+BET2*VIRL
DADT(1)=EXP(TEMP)
DADT(2)=EXP(TEMP)
$ERROR
CMHZ=BSHZ*A(1)
HZLA=BSHZ*A(2)
IF (DVID.EQ.1) THEN ; DV=Viral Load
IPRE=INTERC+SLOPE*(TIME-12)
Y=2*LOG(THETA(7))+( (DV-IPRE)/THETA(7) )**2
ENDIF
IF (DVID.EQ.2 .AND. DV.EQ.0) THEN ; NO dropout
Y=-2*(-CMHZ)
ENDIF
IF (DVID.EQ.2 .AND. DV.EQ.1) THEN ; dropout
Y=-2*(-(CMHZ-HZLA)) - 2*LOG(1 - EXP(-HZLA))
ENDIF
```



A linear disease progress model is used to describe an offset treatment effect.
The dropout Risk (cumulative hazard; CMHZ) is the hazard integral from the start of the study. The incremental Risk (HZLA) is the integral from the last known non-dropout time.
The NM-TRAN abbreviated code used by Hu & Sale illustrates the computation of -2*log likelihood in $ERROR for viral load (DVID.EQ.1) and the dropout indicator (DVID.EQ.2). The dropout indicator has a value of 1 if the patient dropped out before the end of the study.

Slide 14

## Modifications of Hu & Sale Method

- User supplied $SUB CCONTR for joint ELS and LIKE contributions to objective function
- "Standard" code for disease progress prediction



Hu & Sale described a joint model for HIV viral load and drop-out by explicitly computing the -2 log likelihood contribution to the objective function.
An alternative method is to use a user defined CCONTR sub-routine which uses existing NONMEM routines to compute the likelihood.
This alternative method may be simpler to use because it uses 'standard' code for the disease progress model prediction.

Slide 15

## Modified Code

```
$INPUT ID TRT TIME CMT LOCF
DV MDV DVID EVID

$ESTIM MAX=9990 SIG=4 NOABORT
METHOD=CONDITIONAL LAPLACE

$CONTR DATA=(DVID)
$SUBR ADVAN=6 TOL=6
CONTR=contr.for
CCONTR=ccontr_like.for

$MODEL
COMP=(CUMHAZ)
COMP=(HZLAST,INITIALOFF)

$PK
BSHZ=THETA(1) ; Baseline hazard
BETA=THETA(2) ; RD hazard
BET2=THETA(3) ; ID hazard
EFFECT=TRT*THETA(4)
INTRI=(THETA(5)+EFFECT)*EXP(ETA(1))
SLOPI=THETA(6)*EXP(ETA(2))

$DES
DISPRG=INTRI + SLOPI*T
EXPHAZ=EXP(BETA*LOCF + BET2*DISPRG)
DADT(1)=EXPHAZ
DADT(2)=EXPHAZ

$ERROR
CMHZ=BSHZ*A(1) ; Cum hazard overall
HZLA=BSHZ*A(2) ; Cum hazard from last obs
IF (HZLA.LE.0) HZLA=1.0D-10

IF (DVID.EQ.1) THEN
   Y=INTRI + SLOPI*TIME + ERR(1); Status
ENDIF
IF (DVID.EQ.2.AND.DV.EQ.0) THEN
   PD0=EXP(-CMHZ)        ; Pr no dropout
   Y=PD0
ENDIF
IF (DVID.EQ.2.AND.DV.EQ.1) THEN
   PL0=EXP(-(CMHZ-HZLA)) ; Pr no drop last
   PU1=1-EXP(-HZLA)      ; Pr drop unknown
   Y=PL0 * PU1           ; Pr dropout
ENDIF
```



This illustrates how to joint modelling for all 3 models for missingess.
The CRD model is coded by fixing THETA(2) and THETA(3) to zero ($THETA, $OMEGA and $SIGMA records are not shown here)
The RD model is coded by fixing THETA(2) to zero and estimating THETA(1). It uses the LOCF value to predict the hazard during the unobserved period after the last known observation until drop-out is known.
The ID model is coded by fixing THETA(1) to zero. It uses the predicted value of viral load to predict the drop-out hazard.
The $CONTR record identifies the DVID data item to inform the CCONTR sub-routine whether an observation is a viral load value or a drop-out value.
User supplied CONTR and CCONTR sub-routines are defined in the $SUBROUTINE record.
Differential equations are used to integrate the hazard.
Compartment 1 is used to integrate the cumulative hazard since the start of the study.
Compartment 2 is used to integrate the hazard over the period from the last known observation until the time that it is known that dropout has occurred (at some unknown time since the last known observation).
An effect of treatment is assumed to affect the intercept of the disease progress model.

Slide 16

## CCONTR

```
      SUBROUTINE CCONTR (ICALL,CNT,P1,P2,IER1,IER2
      SAVE
C LVR and NO should match values in NSIZES
      PARAMETER(LVR=30,NO=50)
      COMMON /ROCM4/ Y(NO),DATA(NO,3)
      DOUBLE PRECISION CNT,P1,P2,Y
      DIMENSION P1(*),P2(LVR,*)
      TYPE=DATA(1,1)
C Value of TYPE is provided as a user defined data item
      IF (TYPE.EQ.1)THEN
C CELS is used for continuous type data
        CALL CELS(CNT,P1,P2,IER1,IER2)
      ELSE
C CLIK is used for LIKE or -2LL
C first argument is 1 for LIKE and 2 for -2LL
        CALL CLIK(1,CNT,P1,P2,IER1,IER2)
      ENDIF
      RETURN
      END
```



The format of the user supplied CCONTR is shown here. It can be used quite generally. The main user specific feature is to use the value of TYPE (which is determined by a value in the data set) to choose which method should be used t compute the contribution to the likelihood.

Slide 17

# Dropout Data Format

| #ID | TRT | TIME | CMT | LOCF | DV | MDV | DVID | EVID |
|---|---|---|---|---|---|---|---|---|
| 1 | 1 | 0 | 1 | 0 | -0.6 | 0 | 1 | 0 |
| 1 | 1 | 25 | 1 | -0.6 | 28.1 | 0 | 1 | 0 |
| 1 | 1 | 50 | 1 | 28.1 | 53.2 | 0 | 1 | 0 |
| 1 | 1 | 75 | 1 | 53.2 | 81.8 | 0 | 1 | 0 |
| 1 | 1 | 100 | 1 | 81.8 | 108.7 | 0 | 1 | 0 |
| **1** | **1** | **100** | **1** | **108.7** | **0** | **0** | **2** | **0** |
| 2 | 0 | 0 | 1 | 0 | 0.1 | 0 | 1 | 0 |
| 2 | 0 | 25 | 1 | 0.1 | 28.8 | 0 | 1 | 0 |
| 2 | 0 | 25 | 2 | 28.8 | 0 | 1 | 0 | 2 |
| **2** | **0** | **50** | **1** | **28.8** | **1** | **0** | **2** | **0** |

ID 2: Dropout is between time 25 and 50.
Dropout risk compartment is turned on at 25 (CMT=2).



The NONMEM dataset format is shown here.
When DVID is 1 this means the DV observation is of the disease state (e.g. viral load).
When DVID is 2 this means the DV observation is drop-out status (0=not dropped out, 1=dropped out)
The LOCF data item is used when testing the random dropout model.
The CMT data item is set to 2 at the time of the last observation prior to dropout. This turns on this compartment so that it accumulates the hazard of dropping out.
In this example the first subject did not drop-out during the study and the final record has DV=0 to indicate this.
The second subject dropped out between time 25 and 50. A special event record (EVID=2) is used to set the CMT variable to 2 and turn on the second compartment. The final record for this subject indicates that they were known to have dropped out by time 50.

Slide 18

# LIKE vs -2LL Methods
# ID Bias and Imprecision

1000 subjects observed at $t_i$ = 0, 25, 50, 75 and 100
100 replications
Slope 1 u/time SD 1 u
Baseline 0.0001
ID hazard 0.065
Average Dropout 53% (95 percentile 50-56%)




| | LIKE | Success | 79% | 7 h 4 min | -2LL | Success | 44% | 7h 53 min |
|---|---|---|---|---|---|---|---|---|
| | Bias | loCI | hiCI | RMSE | Bias | loCI | hiCI | RMSE |
| Slope | 0.01% | -0.14% | 0.17% | 0.7% | 0.1% | -0.11% | 0.3% | 0.67% |
| PPVslope | -0.15% | -0.77% | 0.46% | 2.7% | -0.23 | -0.77 | 0.32 | 2.64 |
| Baseline | 2.3% | -1.3% | 5.8% | 15.8% | 15% | 11% | 17% | 10% |
| ID hazard | -0.16% | -0.8% | 0.5% | 2.9% | -2.2% | -2.7% | -1.7% | 1.7% |



The influence of using the two different methods (CCONTR and -2LL) for computing the contribution to the likelihood was investigated by Monte Carlo simulation. The probability of being observed to have dropped out is shown using an informative dropout model in the graph.
Overall about 50% of subjects are expected to drop-out during the time from 0 to 100.
A disease progress observation was made on up to 4 occasions after entry to the study.
During simulation the ID model was used to predict if drop-out occurred between the last visit and the current visit. A linear disease progress model with fixed intercept of 0 was used to describe the disease progress state. Simulated data were then fitted to the ID model using NONMEM.
The CCONTR method was faster and slightly more NONMEM runs minimized successfully. The bias and imprecision of the parameter estimates was similar for both methods. There were major biases in the estimated of the baseline and informative dropout hazard parameters.

Slide 19

# CRD (null) Randomization Test

1000 subjects observed at t = 0, 25, 50, 75 and 100
1000 replications; RD and ID: One extra parameter
Bootstrap mean and 95% confidence interval
Slope 1 u/time SD 1 u
Baseline 0.01
ID hazard 0.0
Average Dropout 50% (95 percentile 47-53%)
Type I error rate 5%




| Model Comparison | CritOBJ | Low | High | F success |
|---|---|---|---|---|
| Null: CRD Alternate: RD | 3.75 | 3.27 | 4.25 | 95% |
| Null: CRD Alternate: ID | 3.67 | 3.33 | 4.17 | 95% |



Hu and Sale investigated the three models for missingness using real data sets. They distinguished between the models by assuming the difference in -2 times the log likelihood was distributed according to the chi-square distribution.
This assumption was tested by simulating data under the completely random dropout model and then fitting it with the same model and with the random and informative dropout models.
The probability of being observed to have dropped out is shown using the CRD model in the graph. Overall about 50% of subjects are expected to drop-out during the time from 0 to 100.
The addition of one extra parameter with either the RD or ID models required a large change in objective function to reject the null with a 5% Type I error rate. Over 90% of runs minimized successfully with both the null model (CRD) and the alternate model (RD or ID).

Slide 20

# Questions

- Is Missingness Informative?
  - Only one bit of information per subject
- Can NONMEM get the right answer?
  - LIKE method with CCONTR is OK
  - Direct coding of -2LL is biased
- Can we distinguish CR, RD and ID?
  - Randomization test shows ΔOBJ is approximately $\chi^2$ distributed



Missingness models can be used to simulate drop-outs but the data from a drop-out is small and is unlikely to help identify the underlying disease progress model because each subject only provides one bit of information.
NONMEM estimates of the drop-out hazard are biased and it seems unlikely that they are reliable for simulation the drop-out process.
The very large change in objective function required to reject the null is unexpected and raises questions about the implementation of the model and or NONMEMs methods.

Slide 21

# More Material

From
John Abowd

Slide 22

## Missing Data Mechanisms

- The complete data are defined as the matrix $Y$ ($n \times K$).
- The **pattern** of missing data is summarized by a matrix of indicator variables $M$ ($n \times K$).
- The data generating **mechanism** is summarized by the joint distribution of $Y$ and $M$.

$$m_{ij} = \begin{cases} 0, \text{if } y_{ij} \text{ is observed} \\ 1, \text{if } y_{ij} \text{ is missing} \end{cases}$$

$$p\left(Y, M \middle| \theta, \phi\right)$$

From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8



Slide 23

## Missing Completely at Random

- In this case the missing data mechanism does not depend upon the data $Y$.
- This case is called MCAR.

$$p(M|Y, \theta, \phi) = p\left(M \middle| \phi\right)$$

From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8



Slide 24

## Missing at Random

- Partition Y into observed and unobserved parts.
- Missing at random means that the distribution of M depends only on the observed parts of Y.
- Called MAR.

$$Y = \left(Y_{\text{obs}}, Y_{\text{mis}}\right)$$

$$p(M|Y, \theta, \phi) = p(M|Y_{\text{obs}}, \phi)$$

From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8

| | | |
|---|---|---|
| Slide 25 | **Not Missing at Random**<br>• If the condition for MAR fails, then we say that the data are not missing at random, NMAR.<br>• E.g. dropout because of adverse effects or failure of drug to be effective<br>From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8<br> | |
| Slide 26 | **The Rubin and Little Taxonomy for Dealing with Missing Values**<br>• Analysis of the complete records only<br>• Weighting procedures<br>• Imputation-based procedures<br>• Model-based procedures<br>From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8<br> | |
| Slide 27 | **Analysis of Complete Records Only**<br>• Assumes that the data are MCAR.<br>• Only appropriate for small amounts of missing data.<br>• Used to be common in economics, less so in sociology.<br>• Now very rare.<br>From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8<br> | |

Slide 28

## Weighting Procedures

- Modify the design weights to correct for missing records.
- Provide an item weight (e.g., earnings and income weights in the CPS) that corrects for missing data on that variable.

From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8



Slide 29

## Imputation-based Procedures

- Missing values are filled-in and the resulting "Completed" data are analyzed
  - Mean imputation
  - Regression imputation
- Some imputation procedures (e.g., Rubin's multiple imputation) are really model-based procedures.

From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8



Slide 30

## Model-based Procedures

- A probability model based on $p(Y, M)$ forms the basis for the analysis.
- This probability model is used as the basis for estimation of parameters or effects of interest.
- Some general-purpose model-based procedures are designed to be combined with likelihood functions that are not specified in advance.

From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8

Slide 31

## Little and Rubin's Principles

- Imputations should be
  - Conditioned on observed variables
  - Multivariate
  - Draws from a predictive distribution
- Single imputation methods do not provide a means to correct standard errors for estimation error.

From John Abowd http://instruct1.cit.cornell.edu/courses/cis440/8



Slide 32

## Hazard Models for Dropout

CRD: $h(t; Y_O, Y_U, \eta, \beta) = \beta_0$

RD: $h(t; Y_O, Y_U, \eta, \beta) = \beta_0 \exp(\beta_1 Y_{i-1})$, for $t_{i-1} \le t < t_i$

ID: $h(t; Y_O, Y_U, \eta, \beta) = \beta_0 \exp(\beta_2 Y_U)$, for $t_{i-1} \le t < t_i$

$Y_{i-1}$ means LOCF
$Y_U$ means disease progress model prediction

 Hsu C, Sale ME. A joint model for nonlinear longitudinal data with informative dropout. J Pharmacokinet Pharmacodyn 2003;30(1):83-103

Three models for the hazard are shown. Each one corresponds to the missingness mechanism for dropout.
CRD is completely random dropout. The hazard is constant and is not affected by any observed or unobserved value of disease state.
RD is random dropout. The hazard depends on the last observed value of the disease state. The last value is carried forward to the current time and therefore it is called a LOCF or Last Observation Carried Forward value.
ID is informative dropout. The hazard depends on the predicted value for the disease state.

Slide 33

## A Special Case

```
$ERROR
CMHZ=BSHZ*A(1) ; Cum hazard overall
HZLA=BSHZ*A(2) ; Cum hazard from last obs
IF (HZLA.LE.0) HZLA=1.0D-10

IF (DVID.EQ.1) THEN
   Y=INTRI + SLOPI*TIME + ERR(1); Status
ENDIF
IF (DVID.EQ.2.AND.DV.EQ.0) THEN
   PD0=EXP(-CMHZ)         ; Pr no dropout
   Y=PD0
ENDIF
IF (DVID.EQ.2.AND.DV.EQ.1) THEN
   PL0=EXP(-(CMHZ-HZLA)) ; Pr no drop last
   PU1=1-EXP(-HZLA)       ; Pr drop unknown
   Y=PL0 * PU1            ; Pr dropout
ENDIF
; Dropout at time of visit
IF (DVID.EQ.2.AND.DV.EQ.2) THEN
   PL0=EXP(-(CMHZ-HZLA)) ; Pr no drop last
   PU1=THETA(PVISIT)      ; Pr due to visit
   Y=PL0 * PU1            ; Pr dropout
ENDIF
```



The probability of dropping out may also be affected by the process of making a visit to see the clinician. The clinician may advise the patient to withdraw at the time of the visit. This might be modelled by having a third category for the drop-out variable. When DV is 2 this indicates the patient dropped out at that time. The probability of this happening is described by an additional probability parameter (THETA(PVISIT)).